# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=intersect
_pkgname=Intersect
pkgver=2.5
pkgrel=5
pkgdesc='Post-exploitation framework.'
arch=('x86_64' 'armv7h' 'aarch64')
url="https://github.com/ohdae/Intersect-2.5"
license=('custom')
groups=('blackarch' 'blackarch-automation')
depends=('python')
makedepends=('git')
source=("git+https://www.github.com/ohdae/$_pkgname-$pkgver.git"
        "$_pkgname.patch")
sha512sums=('SKIP'
            '280e5a1590f5c02b3b71d0ce66a480742f628b47ac0a884e3d16069db90ece1f8162aca0e37d09fe43a352ae2372d6b79e790e84e48436657c6f53ae763bc80e')

prepare() {
  cd "$_pkgname-$pkgver"

  patch -p1 -i "$srcdir/Intersect.patch"
}

package() {
  cd "$_pkgname-$pkgver"

  install -dm 755 "$pkgdir/usr/share/Intersect"

  install -Dm 755 Create.py "$pkgdir/usr/bin/$pkgname"

  for i in Docs/ Junk/ Logs/ Scripts/ src/ Storage/ Tools/ ; do
    cp -R $i "$pkgdir/usr/share/Intersect/";
  done
}

